The higher-derivative example from section 2.4 of hep-th/0701238

The identity which we will prove here comes from appendix A of `hep-th/0111128`. The declaration of the indices, Weyl tensor and covariant derivative:

```
{i,j,m,n,k,p,q,l,r,r#}::Indices(vector).
C_{m n p q}::WeylTensor.
\nabla{#}::Derivative.
\nabla_{r}{ C_{m n p q} }::SatisfiesBianchi.
```

Assigning list property Indices to $\{i,\ j,\ m,\ n,\ k,\ p,\ q,\ l,\ r,\ r\#\}$.
Assigning property WeylTensor to C_{mnpq}.
Assigning property Derivative to $\nabla\#$.
Assigning property SatisfiesBianchi to $\nabla_r C_{mnpq}$.
The identity which we want to prove:

```
Eij:=- C_{i m k l} C_{j p k q} C_{l p m q} + 1/4 C_{i m k l} C_{j m p q} C_{k l p q}
     - 1/2 C_{i k j l} C_{k m p q} C_{l m p q}:

E:=  C_{j m n k} C_{m p q n} C_{p j k q} + 1/2 C_{j k m n} C_{p q m n} C_{j k p q}:

exp:= \nabla_{i}{\nabla_{j}{ @(Eij) }} - 1/6 \nabla_{i}{\nabla_{i}{ @(E) }};
```

$$\begin{aligned} exp\ :=\ & \nabla_i\nabla_j(-C_{imkl}C_{jpkq}C_{lpmq}+\frac{1}{4}\,C_{imkl}C_{jmpq}C_{klpq}-\frac{1}{2}\,C_{ikjl}C_{kmpq}C_{lmpq}) \\ & -\frac{1}{6}\,\nabla_i\nabla_i(C_{jmnk}C_{mpqn}C_{pjkq}+\frac{1}{2}\,C_{jkmn}C_{pqmn}C_{jkpq}); \end{aligned}$$

First apply the product rule to write out the derivatives,

```
@distribute!(%): @prodrule!(%):
@distribute!(%): @prodrule!(%):

@prodsort!(%): @canonicalise!(%): @rename_dummies!(%): @sumflatten!(%):
@collect_terms!(%);
```

$$
\begin{aligned}
exp \; := \; & C_{ijmn} C_{ikmp} \nabla_q \nabla_j C_{nkpq} - C_{ijmn} \nabla_k C_{ipmq} \nabla_p C_{jqnk} - 2\, C_{ijmn} \nabla_i C_{mkpq} \nabla_p C_{jknq} \\
& - C_{ijmn} \nabla_k C_{ikmp} \nabla_q C_{jpnq} + C_{ijmn} C_{ikmp} \nabla_j \nabla_q C_{nkpq} - 2\, C_{ijmn} \nabla_i C_{jkmp} \nabla_q C_{nkpq} \\
& - C_{ijmn} C_{ikpq} \nabla_m \nabla_p C_{jqnk} - \frac{1}{4}\, C_{ijmn} C_{ijkp} \nabla_q \nabla_m C_{nqkp} + \frac{1}{4}\, C_{ijmn} \nabla_k C_{ijpq} \nabla_p C_{mnkq} \\
& - \frac{1}{2}\, C_{ijmn} \nabla_i C_{jkpq} \nabla_k C_{mnpq} - \frac{1}{4}\, C_{ijmn} \nabla_k C_{ijkp} \nabla_q C_{mnpq} \\
& - \frac{1}{4}\, C_{ijmn} C_{ijkp} \nabla_m \nabla_q C_{nqkp} - \frac{1}{2}\, C_{ijmn} \nabla_i C_{mnkp} \nabla_q C_{jqkp} + \frac{1}{4}\, C_{ijmn} C_{ikpq} \nabla_j \nabla_k C_{mnpq} \\
& - \frac{1}{2}\, C_{ijmn} C_{ijmk} \nabla_p \nabla_q C_{npkq} + C_{ijmn} \nabla_k C_{ijmp} \nabla_q C_{nqkp} - C_{ijmn} \nabla_k C_{ijmp} \nabla_q C_{nkpq} \\
& + \frac{1}{2}\, C_{ijmn} C_{ikpq} \nabla_m \nabla_j C_{nkpq} + \frac{1}{2}\, C_{ijmn} \nabla_i C_{mkpq} \nabla_n C_{jkpq} - \frac{1}{2}\, C_{ijmn} \nabla_i C_{jkpq} \nabla_m C_{nkpq} \\
& + \frac{1}{2}\, C_{ijmn} C_{ikpq} \nabla_j \nabla_m C_{nkpq} + \frac{1}{2}\, C_{ijmn} C_{ikmp} \nabla_q \nabla_q C_{jknp} + C_{ijmn} \nabla_k C_{ipmq} \nabla_k C_{jpnq} \\
& - \frac{1}{4}\, C_{ijmn} C_{ijkp} \nabla_q \nabla_q C_{mnkp} - \frac{1}{2}\, C_{ijmn} \nabla_k C_{ijpq} \nabla_k C_{mnpq};
\end{aligned}
$$

Because the identity which we intend to prove is only supposed to hold on Einstein spaces, we set the divergence of the Weyl tensor to zero,

```
@substitute!(%)( \nabla_{i}{C_{k i l m}} -> 0, \nabla_{i}{C_{k m l i}} -> 0 );
```

$$
\begin{aligned}
exp \; := \; & C_{ijmn} C_{ikmp} \nabla_q \nabla_j C_{nkpq} - C_{ijmn} \nabla_k C_{ipmq} \nabla_p C_{jqnk} - 2\, C_{ijmn} \nabla_i C_{mkpq} \nabla_p C_{jknq} \\
& - C_{ijmn} C_{ikpq} \nabla_m \nabla_p C_{jqnk} - \frac{1}{4}\, C_{ijmn} C_{ijkp} \nabla_q \nabla_m C_{nqkp} + \frac{1}{4}\, C_{ijmn} \nabla_k C_{ijpq} \nabla_p C_{mnkq} \\
& - \frac{1}{2}\, C_{ijmn} \nabla_i C_{jkpq} \nabla_k C_{mnpq} + \frac{1}{4}\, C_{ijmn} C_{ikpq} \nabla_j \nabla_k C_{mnpq} \\
& + \frac{1}{2}\, C_{ijmn} C_{ikpq} \nabla_m \nabla_j C_{nkpq} + \frac{1}{2}\, C_{ijmn} \nabla_i C_{mkpq} \nabla_n C_{jkpq} - \frac{1}{2}\, C_{ijmn} \nabla_i C_{jkpq} \nabla_m C_{nkpq} \\
& + \frac{1}{2}\, C_{ijmn} C_{ikpq} \nabla_j \nabla_m C_{nkpq} + \frac{1}{2}\, C_{ijmn} C_{ikmp} \nabla_q \nabla_q C_{jknp} + C_{ijmn} \nabla_k C_{ipmq} \nabla_k C_{jpnq} \\
& - \frac{1}{4}\, C_{ijmn} C_{ijkp} \nabla_q \nabla_q C_{mnkp} - \frac{1}{2}\, C_{ijmn} \nabla_k C_{ijpq} \nabla_k C_{mnpq};
\end{aligned}
$$

This expression should vanish upon use of the Bianchi identity. By expanding all tensors using their Young projectors, this becomes manifest,

```
@young_project_product!(%):
@sumflatten!(%):
@canonicalise!(%):
@rename_dummies!(%):
@collect_terms!(%);
```

@canonicalise: not applicable.

$$exp \; := 0;$$

This proves the identity.